

An Introduction to ODS



Neville Cope
Amgen EU Biostatistics

What is the ODS?

The ODS, or Output Delivery System was the biggest single change to the SAS system in Version 7. The ODS has many features and options, but this poster looks at generating SAS output using the ODS. Previously, SAS output was stored in a simple text file without any formatting options of any kind. Columns were aligned with spaces and tabs and you were very limited in the way you could lay out the output- Even making tables with solid lines proved to be a headache!

PROC TABULATE without use of ODS

		Sex	
		F	M
Height	N	9	10
	Mean	60.59	63.91
	StdDev	5.02	4.94
Weight	N	9	10
	Mean	90.11	108.95
	StdDev	19.38	22.73
Age	N	9	10
	Mean	13.22	13.40
	StdDev	1.39	1.65

In this modern age of word processors and the internet, users wanted more control over their output.

To get SAS 'flat text' output into these applications involved many number of manual tasks, DDE calls, VB Macros, etc...

However, the ODS allows output to be created and formatted for a variety of different 'destinations' including SAS datasets, HTML for the web, WORD-compatible RTF files, high-resolution printer files and more recently Adobe PDF documents.

PROC TEMPLATE

The 'template' for your output needs to only be created once, and is used by most of the output destinations. This means that if you use proc report, proc tabulate or even proc print to create RTF, HTML or PDF documents the outputs will all look consistent, with the same colours, fonts, etc...

The SAS System comes shipped with a lot of templates already predefined:

PROC TABULATE with Default template

PROC TABULATE with Striped template

PROC TABULATE with BarrethBlue template

PROC TABULATE with Strick template

As you can see, these templates were really designed for use with the first ODS destination to be implemented - HTML - and all these colours and fancy formatting are rarely applied to clinical study report output (data listings, summary tables and graphs)

However, new templates can be created using PROC TEMPLATE, but unfortunately an introduction to this is well beyond the scope of this poster!

Simple Templates

At AMGEN, we have a set of standard templates already defined, in a more simple black & white style, with separate templates for various orientations and font sizes

PROC TABULATE with 'Amgen.Styles.PorraitA4FS9' template

		Sex	
		F	M
Height	N	9	10
	Mean	60.59	63.91
	StdDev	5.02	4.94
Weight	N	9	10
	Mean	90.11	108.95
	StdDev	19.38	22.73
Age	N	9	10
	Mean	13.22	13.40
	StdDev	1.39	1.65

Creating Output

So - now you've seen what some ODS output looks like, how do you go about creating some?

SAS can create files that MS Word can read using the RTF (Rich Text Format) destination by simply enclosing the output procedure in ODS RTF statements.

The example above using the Amgen template was created using the following code:

```
ods rtf file="&outdir.\ex3.rtf" style=Amgen.Styles.PorraitA4FS9;
proc tabulate data=sashelp.class;
title "PROC TABULATE with 'Amgen.Styles.PorraitA4FS9' template";
var height weight age;
class sex;
tables (height weight age)*(N MEAN STDDEV),sex;
run;
ods rtf close;
```

Twinking The Output

Once you have code in place to create an RTF file, you can begin to customise it in many different ways.

In the examples above, when the PROC TABULATE is run the ODS writes all the commands that make up an RTF file out to the specified file. In doing so it references the TEMPLATE you have chosen (for fonts and colours, etc), and creates basic layouts for the columns. Some of the formatting, such as exact column widths and page breaking, are actually left to the word processor to sort out.

These can all be overridden by the use of the ODS ESCAPECHAR option. This sets a character, such as "~", which when encountered while creating the output, stops SAS writing RTF code to a file and instead writes out a line of commands you specify.

This functionality allows you to fine-tune your output with things like manual line breaks, bolding, underlining, "traffic lighting", indentation, etc...

Formatting with STYLE

One way to add formatting to your output is using Style Definitions. These are predefined keywords that work across most ODS destinations and allow you to easily change particular attributes of a cell in your output. You can also do this dynamically, by telling SAS to apply these styles based on a format applied to that cell. This is accomplished by using the ESCAPECHAR followed by S (which stands for Style) like this:

"~S={style_element=xxx}"

Where *style_element* is the style to change and *xxx* is the value you want the parameter to take (or the name of the format to use).

In the example below, the name of all the males is formatted with bold characters, and weights of 100 or more are coloured red.

```
proc format;
value hvy low-<100 = 'BLACK'
100-high = 'RED';
run;

ods escapechar='~';

data class2;
set sashelp.class;
length name2 $50;
if sex='M' then name2=~S={font_weight=bold} "||name;
else name2=name;
run;

ods rtf file="&outdir.\ex4.rtf"
style=Amgen.Styles.PorraitA4FS9;

proc report data=class2 nowd spacing=2;
title "PROC REPORT with ODS ESCAPECHAR";
column name2 age height weight;
define name2 / display "Name";
define age / display "Age" format=3.;
define height / display "Height" format=5.1;
define weight / display "Weight" style={foreground=hvy.};
run;

ods rtf close;
```

PROC REPORT with ODS ESCAPECHAR

Name	Age	Height	Weight
Alfred	14	69.0	112.5
Alice	13	56.5	84
Barbara	13	65.3	98
Carol	14	62.8	102.5
Henry	14	63.5	102.5
James	12	57.3	83
Jane	12	59.8	84.5
Janet	15	62.5	112.5
Jeffrey	13	62.5	84
John	12	59.0	99.5
Joyce	11	51.3	50.5
Judy	14	64.3	90
Louise	12	56.3	77
Mary	15	66.5	112
Philip	16	72.0	150
Robert	12	64.8	128
Ronald	15	67.0	133
Thomas	11	57.5	85
William	15	66.5	112

Advanced Style Features

Using various other bits of style formatting code, some pretty advanced formatting can be achieved.

This includes footnotes containing "Page X of Y", font changes, superscript and subscript and even inserting equations using MS field codes:

```
data test;
text="Chi-square symbol: ~S={font_face=symbol} c~(super 2)~S={} using
style definitions";
output;
text="Normal ~{super superscript} ~{sub subscript}";
output;
text="{Cube root using MS Equations: \field{\*fldinst {EQ \R (3,27)
}}";
output;
run;

ods rtf file="&outdir.\ex5.rtf" style=Amgen.Styles.PorraitA4FS9;

title "Advanced Style Options";
footnote;
proc print data=test style={protectspecialchars=off};
run;

ods rtf close;
```

Advanced Style Options

- | Obs | text |
|-----|---|
| 1 | Chi-square symbol: χ^2 using style definitions |
| 2 | Normal ^{superscript} _{subscript} |
| 3 | Cube root using MS Equations: $\sqrt[3]{27}$ |

Complete Control

If you want to take complete control over your output, then instead of using the pre-defined style elements you can force SAS to write raw code to open ODS destinations. This is reasonably easy for the HTML destination, but raw RTF code is more complicated.

Copies of the RTF file definition are freely available off the internet for download. However, a lot of the raw RTF code cannot be used with the ODS (e.g. code to create tables or control page layout, because SAS would have already written these commands to the output file) but table cell and paragraph formatting is available.


The use of this 'raw code' feature is similar to using Style Definitions, but instead the keyword is R (for Raw), followed by the RTF commands enclosed in single quotes. Some examples are presented below:

Using RAW RTF Code

Text	Result
-R'\b ' bold-R'\b0 ' text	Bold text
-R'\ul ' Underlined-R'\ul0 ' text	<u>Underlined text</u>
-R\q ' Centred text	Centred text
-R\qr ' Right aligned text	Right aligned text
-R\l300 ' Indented text	Indented text
manual-R'\line 'Line-R'\line 'Breaks	manual Line Breaks
-R'\tdcc\cx600 ' 10000.9	10000.9 (decimals aligned)
-R'\tdcc\cx600 ' 1.9	1.9 (decimals aligned)
102 -R'\ucl\u8005 ' 100 (special characters)	102 > 100 (special characters)

Graphics

As well as sending PROC REPORT output directly to an RTF file, the ODS also allows graphics to be created using the same syntax. This negates the need to manually insert CGM/JPG/GIF graphics files into MS Word, and with the correct options can allow multiple plots to be contained in one document.



Conclusions

Using the ODS may seem daunting initially, and finding which options are required to generate the exact output you want can be a trial-and-error process in the beginning, but overall it is a very powerful tool for reporting your clinical study data and I feel the benefits far outweigh the limitations.

It became a production feature in version 8.1 of SAS, and has been continuously updated and improved ever since. Version 9.1 promises even more powerful features and updates, including more control over borders in tables and the new ODS DOCUMENT destination.

This poster has aimed to give an overview of the ODS and the RTF destination. More information can be found at the SAS Support website: <http://support.sas.com/rnd/base/topics/odsrtf/>

If you require further examples or example code, or more information on the topics discussed here, please contact me via email: nevillec@amgen.com